
1. Modellbildung und Simulation 

 

Im folgenden Kapitel wird die symmetrische Hälfte der Rissprobe modelliert und simuliert. 

 

1.1 Kopf des Ansys-Dokuments 

 

fini 

/clear 

/prep7 

 

1.2 Parameter 

 

a/W-Werte (0.25, 0.3, 0.4, 0.5, 0.6, 0.7) 

 

aW=0.7   ! a/W Wert 

 

Zu verschiebende Mittelpunkte (zwecks Singularität) 

 

 
Abbildung 1 Singularität (Knotennummerierung Bsp. a/W = 0.7) 

 

 

nl=274   ! Mittelpunkt links unten 

nr=273   ! Mittelpunkt rechts unten 

nm=668   ! Mittelpunkt mitte 

 

Restliche Parameter 

 

pi=3.141592654 

l=250    ! Länge Platte (x-Länge) 

b=120    ! Breite Platte (y-Länge) 

W=200    ! Position Bohrung in X-Richtung 

h=55    ! Position Bohrung in Y-Richtung 

d=50     ! Durchmesser Bohrung 

delta=100   ! Dicke der Probe 

kl=5    ! Kantenlänge eines Elements 

r=l-W+(W*aW)  ! Risslänge 

 

1.3 Elementdefinition 

 

Verwendet werden „Scheibenelemente“ (Plane 82) mit 4 Eckpunkten bzw. 4 Kanten sowie 4 

Mittelpunkten (also insgesamt 8 Punkte). Die Dicke der Scheibe muss zusätzlich angegeben werden. 

 

ET,1,82,0,0,3 

R,1,delta 

 

1.4 Materialdefinition 

 

Von der Aufgabenstellung sind 4 Materialien vorgegeben, welche für jedes a/W-Verhältnis 

betrachtet werden sollen. Das erste Material (nachfolgend wird von M1 bis M4 durchgezählt) ist 

isotrop, die restlichen Materialien sind orthotrop. 

 

mp,ex,1,31200000  ! E-Modul in x-Richtung (in kg/(s²*mm)) 

mp,ey,1,31200000  ! E-Modul in y-Richtung (in kg/(s²*mm)) 

mp,ez,1,31200000  ! E-Modul in z-Richtung (in kg/(s²*mm)) 

mp,nuxy,1,0.3  ! Querkontraktionszahl in x-y-Richtung 

mp,gxy,1,12000000 ! Schubmodul in x-y-Richtung 

 

!mp,ex,1,40950000 ! E-Modul in x-Richtung (in kg/(s²*mm)) 

!mp,ey,1,13000000 ! E-Modul in y-Richtung (in kg/(s²*mm)) 

!mp,ez,1,40950000 ! E-Modul in z-Richtung (in kg/(s²*mm)) 

!mp,nuxy,1,0.3  ! Querkontraktionszahl in x-y-Richtung 

!mp,gxy,1,5900000 ! Schubmodul in x-y-Richtung 

 

!mp,ex,1,46100000 ! E-Modul in x-Richtung (in kg/(s²*mm)) 

!mp,ey,1,18800000 ! E-Modul in y-Richtung (in kg/(s²*mm)) 

!mp,ez,1,46100000 ! E-Modul in z-Richtung (in kg/(s²*mm)) 

!mp,nuxy,1,0.3  ! Querkontraktionszahl in x-y-Richtung 

!mp,gxy,1,9350000 ! Schubmodul in x-y-Richtung 

 

!mp,ex,1,51300000 ! E-Modul in x-Richtung (in kg/(s²*mm)) 

!mp,ey,1,24500000 ! E-Modul in y-Richtung (in kg/(s²*mm)) 

!mp,ez,1,51300000 ! E-Modul in z-Richtung (in kg/(s²*mm)) 

!mp,nuxy,1,0.3  ! Querkontraktionszahl in x-y-Richtung 

!mp,gxy,1,8800000 ! Schubmodul in x-y-Richtung 

 

1.5. Geometrie 

 

Zunächst werden alle benötigten Punkte (Keypoints) gesetzt. Dies sind die 4 Eckpunkte, 2 Punkte auf 

dem Umfang der Bohrung sowie die Rissspitze (zur leichteren Identifikation im späteren Knotenbild). 

k,1,0,0 

k,2,0,b 

k,3,l,b 

k,4,l,0 

k,5,l-r,0 

k,6,W-(d/2),h 

k,7,W+(d/2),h 

 

Im Anschluss werden die Punkte mit Linien verbunden. Die Kreisbögen (2 x 180°) werden mit larc 

definiert. 

 

l,1,2 

l,2,3 

l,3,4 

l,4,5 

l,5,1 

larc,6,7,2,(d/2) 

larc,6,7,4,(d/2) 

 

Danach wird zunächst die Rechteckfläche der Probe, dann die Kreisfläche der Bohrung erstellt. Im 

Anschluss werden beide Flächen subtrahiert. 

 

AL,1,2,3,4,5 

AL,6,7 

ASBA,1,2 

 

1.6 Vernetzung 

 

Die aus der Subtraktion resultierende Fläche (Fläche 3) wird automatisch vernetzt. Die Elementgröße 

wird dabei mit der vorgegebenen Kantenlänge kl definiert. 

 

AESIZE,3,kl 

AMESH,3 

 

1.7 Singularität an der Rissspitze 

 

Um die entstehende Singularität an der Rissspitze nachzubilden, werden die Kantenmittelpunkte um 

die Rissspitze zur Rissspitze hin verschoben. Dabei muss das Verhältnis ¼ zu ¾ eingehalten werden. 

Abbildung 1 zeigt die Verschiebung. 

 

Im ersten Schritt müssen die zu verschiebenden Mittelpunkte selektiert werden. 

 

NSEL,S,NODE,,nr,nl,1 

NSEL,A,NODE,,nm,nm,1 

Im nächsten Schritt wird die geometrische Verknüpfung dieser Mittelpunkte mit dem Netz 

aufgehoben. Dies ist Voraussetzung zum Verschieben. 

 

Modmsh,nocheck 

 

Im letzten Schritt werden die 3 Mittelpunkte jeweils um ¼ der Kantenlänge kl in Richtung Rissspitze 

verschoben. Danach wird die Selektion aufgehoben. 

 

n,nl,nx(nl)+(kl/4),ny(nl),nz(nl) 

n,nr,nx(nr)-(kl/4),ny(nr),nz(nr) 

n,nm,nx(nm),ny(nm)-(kl/4),nz(nm) 

NSEL,ALL 

 

1.8 Einspannungen 

 

Die Rissspitze wird fest eingespannt, um zumindest einen Knoten des Netzes in x-Richtung zu 

sperren. Durch die Symmetrie der Probe und somit der Belastung werden alle restlichen Knoten 

entlang der Symmetrielinie bis zur linken unteren Ecke der Probe in y-Richtung gesperrt. 

 

D,198,ALL 

D,nl,UY,0,,296,1 

D,1,UY,0 

 

1.9 Belastung 

 

Auf den oberen Punkt des Kreisumfangs der Bohrung wirkt eine Kraft in y-Richtung (nach oben). 

Diese wurde mit 1000N festgelegt. 

 

F,345,FY,1000 

 

1.10 Lösen 

 

Mit den folgenden Befehlen wird das Modell (siehe Abbildung) gelöst und der Postprozessor 

gestartet. 

 

/solu 

solve 

/post1 

pldisp,1 

Abbildung 2 Vernetzung (Darstellung aller Knoten, Einspannungen, Kräfte, Bsp. für a/W = 0.7) 

 

 

Es wird die verformte Probe dargestellt (siehe folgende Abbildungen). 

 

 
Abbildung 3 Verformung (Bsp. für a/W = 0.7) 

 

 

2. Ergebnisse und Auswertung 

 

Für alle Materialien und a/W-Verhältnisse wird die y-Verschiebung des unteren, rechten Eckpunkts 

der Probe abgelesen. Im Anschluss werden die COD-Werte berechnet. Mit Maple werden schließlich 

die Regressionskurven mit Hilfe des Kleinste-Quadrate-Verfahrens bestimmt und dargestellt. 

2.1 y-Verschiebung 

 
Tabelle 1 y-Verschiebungswerte aller a/W-Verhältnisse und Materialien 

a/W M1 M2 M3 M4 

0,25 3,3396E-06 5,2985E-06 3,7739E-06 3,4387E-06 

0,3 3,9784E-06 6,3293E-06 4,4950E-06 4,1079E-06 

0,4 5,7843E-06 9,3381E-06 6,5997E-06 6,0304E-06 

0,5 8,7587E-06 1,4580E-05 1,0264E-05 9,2972E-06 

0,6 1,4244E-05 2,4764E-05 1,7372E-05 1,5517E-05 

0,7 2,6474E-05 4,8223E-05 3,3727E-05 2,9683E-05 

 

2.2 COD-Werte 

 

Die COD-Werte berechnen sich mit folgender Formel 

 

           , 

 

wobei    die y-Verschiebung,    der E-Modul in y-Richtung sowie   die (Scheiben-)Dicke der Probe 

ist. 

 
Tabelle 2 COD-Werte aller a/W-Verhältnisse und Materialien 

a/W M1 M2 M3 M4 

0,25 10419,552 6888,05 7094,932 8424,815 

0,3 12412,608 8228,09 8450,6 10064,355 

0,4 18047,016 12139,53 12407,436 14774,48 

0,5 27327,144 18954 19296,32 22778,14 

0,6 44441,28 32193,2 32659,36 38016,65 

0,7 82598,88 62689,9 63406,76 72723,35 

Delta (mm) 100 100 100 100 

EY (kg/(s²*mm)) 31200000 13000000 18800000 24500000 

 

2.3 Regressionskurven 

 

Kopf des Maple-Dokuments 

 

>restart; 

>with(Statistics):with(CurveFitting):with(plots): 

 

Eingabe aller Daten 

 

>aW_Werte:=[0.25,0.3,0.4,0.5,0.6,0.7]: 

COD_M1:=[10419.552,12412.608,18047.016,27327.144,44441.28,82598.88]: 

COD_M2:=[6888.05,8228.09,12139.53,18954,32193.2,62689.9]: 

COD_M3:=[7094.932,8450.6,12407.436,19296.32,32659.36,63406.76]: 

COD_M4:=[8424.815,10064.355,14774.48,22778.14,38016.65,72723.35]: 

 

Erstellen der Punkte-Kurven 

 

>Punkte_M1:=ScatterPlot(aW_Werte,COD_M1,color=blue, 

symbol=box,labels=["a/W","COD"]): 

Punkte_M2:=ScatterPlot(aW_Werte,COD_M2,color=red, 

symbol=cross,labels=["a/W","COD"]): 

Punkte_M3:=ScatterPlot(aW_Werte,COD_M3,color=green, 

symbol=circle,labels=["a/W","COD"]): 

Punkte_M4:=ScatterPlot(aW_Werte,COD_M4,color="DarkOrchid", 

symbol=cross,labels=["a/W","COD"]): 

 

Erstellen der Regressionskurvenfunktionen mit Hilfe des Kleinste-Quadrate-Verfahrens 

 

>y_M1:=LeastSquares(aW_Werte,COD_M1,aW,curve=a*aW^2+b*aW+c): 

y_M2:=LeastSquares(aW_Werte,COD_M2,aW,curve=a*aW^2+b*aW+c): 

y_M3:=LeastSquares(aW_Werte,COD_M3,aW,curve=a*aW^2+b*aW+c): 

y_M4:=LeastSquares(aW_Werte,COD_M4,aW,curve=a*aW^2+b*aW+c): 

y_M1; 

y_M2; 

y_M3; 

y_M4; 

 

                                        
 

 
                          

 

 
 
 

 

                                        
 

 
                         

 

 
 
 

 

                                        
 

 
                          

 

 
 
 

 

                                        
 

 
                          

 

 
 
 

 

 

Erstellen der Regressionskurven (gerundet) 

 

>regr_M1(aW):=evalf(y_M1,3); 

regcurve_M1:=plot(regr_M1(aW),aW=0.25..0.7,color=blue): 

regr_M2(aW):=evalf(y_M2,3); 

regcurve_M2:=plot(regr_M2(aW),aW=0.25..0.7,color=red): 

regr_M3(aW):=evalf(y_M3,3); 

regcurve_M3:=plot(regr_M3(aW),aW=0.25..0.7,color=green): 

regr_M4(aW):=evalf(y_M4,3); 

regcurve_M4:=plot(regr_M4(aW),aW=0.25..0.7,color="DarkOrchid"): 

Anzeigen der Kurve 

 

>display([Punkte_M1,Punkte_M2,Punkte_M3,Punkte_M4, 

regcurve_M1,regcurve_M2,regcurve_M3,regcurve_M4]); 

 

 
Abbildung 4 Regressionskurven und Punktekurven 

 

 

Die blaue Kurve (Kästchen) ist die Kurve für Material 1. Die rote Kurve (Kreuz) ist die Kurve für 

Material 2. Die grüne Kurve (Kreis) ist die Kurve für Material 3. Die violette Kurve (Kreuz) ist die Kurve 

für Material 4. 

 

3. Diskussion von verschiedenen Elementgrößen 

 

Im folgenden Kapitel wird die Abhängigkeit des Ergebnisses von der Elementgröße diskutiert. Dabei 

soll der Zustand a/W = 0.5 betrachtet werden. Als Material wird Material 1 (M1) gewählt. 

 

Die Elementgröße ist umgekehrt proportional zur Kantenlänge kl. Als Kantenlängen werden 2mm, 

5mm und 8mm gewählt. 

Die folgenden Abbildungen zeigen die Verformungen der Netze mit den unterschiedlichen 

Elementgrößen. 

 

 
Abbildung 5 Verformung, Elemente mit Kantenlänge kl = 2mm (Bsp. für a/W = 0.5, Material 1) 

 

 

 
Abbildung 6 Verformung, Elemente mit Kantenlänge kl = 5mm (Bsp. für a/W = 0.5, Material 1) 

Abbildung 7 Verformung, Elemente mit Kantenlänge kl = 8mm (Bsp. für a/W = 0.5, Material 1) 

 

 

Die Ergebnisse in Form der y-Verschiebungen des unteren, rechten Eckpunkts werden in folgender 

Tabelle dargestellt. 

 
Tabelle 3 y-Verschiebungswerte mit div. Kantenlängen (Bsp. für a/W = 0.5, Material 1) 

Kantenlänge (mm) M1 

2 8,7588E-06 

5 8,7587E-06 

8 8,7617E-06 

 

Da der reale Wert nicht bekannt ist, können die angenäherten Werte nicht diskutiert werden. Es gilt 

allerdings, dass mit kleinerer Elementgröße die Anzahl an Elementen im selben Modell größer wird. 

Je größer die Anzahl an Elementen, also je feiner die Elemente, desto genauer ist das Ergebnis. Somit 

ergibt sich: je kleiner die Kantenlänge, desto genauer das angenäherte Ergebnis. 


